#===============================================================================
# Copyright 2019 Intel Corporation
#
# Licensed under the Apache License, Version 2.0 (the "License");
# you may not use this file except in compliance with the License.
# You may obtain a copy of the License at
#
#     http://www.apache.org/licenses/LICENSE-2.0
#
# Unless required by applicable law or agreed to in writing, software
# distributed under the License is distributed on an "AS IS" BASIS,
# WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
# See the License for the specific language governing permissions and
# limitations under the License.
#===============================================================================

file(GLOB_RECURSE SOURCES
    ${CMAKE_CURRENT_SOURCE_DIR}/*.h
    ${CMAKE_CURRENT_SOURCE_DIR}/*.hpp
    ${CMAKE_CURRENT_SOURCE_DIR}/*.c
    ${CMAKE_CURRENT_SOURCE_DIR}/*.cpp
    )

file(GLOB_RECURSE CL_SOURCES ${CMAKE_CURRENT_SOURCE_DIR}/*.cl)
foreach(cl_file ${CL_SOURCES})
    get_filename_component(kernel_name ${cl_file} NAME_WE)
    set(cpp_file "${CMAKE_CURRENT_BINARY_DIR}/${kernel_name}_cl.cpp")
    # XXX: incremental build does not work with headers from OpenCL code
    add_custom_command(
        OUTPUT ${cpp_file}
        COMMAND ${CMAKE_COMMAND}
            -DCL_FILE="${cl_file}"
            -DCPP_FILE="${cpp_file}"
            -DCL_INC_DIR="${PROJECT_SOURCE_DIR}/src"
            -P ${PROJECT_SOURCE_DIR}/cmake/gen_ocl_source.cmake
        DEPENDS ${cl_file}
    )
    list(APPEND SOURCES ${cpp_file})
endforeach()

set(OBJ_LIB ${LIB_NAME}_ocl)
add_library(${OBJ_LIB} OBJECT ${SOURCES})
set(${LIB_NAME}_SUB_OBJS ${${LIB_NAME}_SUB_OBJS}
    $<TARGET_OBJECTS:${OBJ_LIB}> PARENT_SCOPE)
